
Application (WP;WPWP;Default;"WPWPUS.WCD")

//*** INITIALIZE HOTDOCS
GetWPData(macroPath;MacroPath!)
RUN(macroPath+"hdinit.wcm")

//*** GET TEMPLATE NAME
GetWPData(OrgPath;Path!)
GetWPData(OrgName;Name!)
OrgFile := OrgPath+OrgName

//*** OPEN COMPONENT FILE R/W
DllLoad(HotdocsDll;"HD3.dll")
DllCall(HotDocsDll;
 "OpenComponent";
 Result:Integer;
 {OrgFile;0}
)

//*** IF SOMETHING IS WRONG, QUIT
IF (Result = 0)
 DLLFree(HotDocsDll)
 QUIT
ENDIF

//*** SAVE TEMPLATE
GetWPData(SelectOn;SelectModeActive!)
IF (SelectOn)

EditCopy()
SelectMode(Off!)

ENDIF
GetWPData(modified;DocumentModified!)
IF (modified)

FileSave()
ENDIF

// *** OPEN NEW DOCUMENT AND INSERT TEMPLATE
FileNew()
inFlags := 0
IF (SelectOn)

EditPaste()
inFlags := 64

ELSE
TYPE("X")
DeleteCharPrevious()
FileRetrieve(OrgFile;Yes!;Insert!)



ENDIF

//*** STORE ORGFILE IN DLL
DllLoad(HotdocsDll;"HD3.dll")
DllCall(HotdocsDll;"PutString";sRes:Integer;{0;OrgFile})
DllCall(HotdocsDll;"PutLong";sRes:Integer;{0;inFlags})
DllFree(HotdocsDll)
//*** RUN ASMDOC
GetWPData(macroPath;MacroPath!)
RUN(macroPath+"hdasmdoc.wcm")

//*** RETRIEVE ASMFLAG FROM DLL
DllLoad(HotdocsDll;"HD3.dll")
DllCall(HotdocsDll;"GetLong";AsmFlag:Integer;{0})
DllCall(HotdocsDLL;"GetString";FieldBuf:AnsiString;{0})
DllFree(HotdocsDll)
IF (AsmFlag <> 1)
 CloseNoSave(No!) 
 Close(No!)
ENDIF
IF (AsmFlag = 2)
 PosDocTop()
 STRLEN(len;FieldBuf)
 IF (len <> 0)
   ONNOTFOUND(NoErr@)
   SearchText(FieldBuf;Forward!;Extended!) //*** FIND ERROR FLD
   LABEL(NoErr@)    
 ENDIF
ENDIF
IF (AsmFlag = 3)
 PosDocBottom()
ENDIF
QUIT


